
# WHO AM I... Mike Zdeb
University@Albany
School of Public Health
Rensselaer, NY

# TEACH INTRODUCTORY (3 CREDIT) AND ADVANCED
(1 CREDIT) SAS CLASSES, INTRODUCTORY
STATISTICS IN-CLASS AND ON-LINE

# WE MAY HAVE THE ONLY STUDENTS FROM
MONGOLIA, KHAZHAKSTAN, UZBEKISTAN, AND
NOVOSIBIRSK (IN SIBERIA) WHO ARE 'SAS
CERTIFIED'

# PAPER (CO-AUTHOR IS ROBERT ALLISON FROM
SAS), SAS CODE, DATA ARE ON THE NESUG WEB
SITE IN THE 2006 "INS & OUTS" SECTION ...

www.nesug.org

AND ON LEX JANSEN'S WEB SITE...

www.lexjansen.com

(GREAT WEB SITE TO ACCESS NESUG, SUGI,
PHARMASUG, ETC. --- ONE STOP SEARCHING)

NESUG 2006 - Urban Legends and SAS 
Stanley Fogleman, Harvard Clinical Research Institute

"SAS DOESN’T DO VERY GOOD GRAPHS In a word:
stupefying (ahurissant ?) ... someone spending 15 minutes
or less with the on-line Enterprise Guide tutorial would
discover that this claim is without merit."

 a brief digression ... in 1500+ papers given at NESUG (1997
through 2007) ... how many times has an author used the
word 'stupefying' ... 5 times, 4 times it's the same author ...

# SAS/GRAPH DOES DO VERY GOOD GRAPHS

# REQUIRES SOME INVESTMENT OF TIME

# EVEN WITH ENTERPRISE GUIDE, WORTH KNOWING
PIECES OF INFORMATION THAT HELP TO PRODUCE
GOOD GRAPHS

# SAS FEATURE WITH ITS OWN MANUAL=COMPEX

PROC SQL, PROC TABULATE, MACROS

# SAS/GRAPH HAS(HAD) TWO MANUALS=COMPLEX**2

MANUAL FOR SAS/GRAPH PROCEDURES (GCHART,
GPLOT, GMAP, G3D, G_ETC.)

MANUAL FOR SAS/GRAPH GLOBAL STATEMENTS
(AXIS, SYMBOL, PATTERN, LEGEND, TITLE,
FOOTNOTE, NOTE, ETC.)

# PLUS...

ANNOTATE, GOPTIONS, OUTPUT (HARDCOPY,
GRAPHICS FILES, ODS)=COMPLEX**3

PLUS "SAS/GRAPH STYLE" GRAPHICS IN OTHER
PROCEDURES (PROC UNIVARIATE AND PROC
BOXPLOT) = COMPLEX**4

ODS STATISTICAL GRAPHICS ADDED TO SAS/STAT
PROCEDURES = COMPLEX**5

# ODS STATISTICAL GRAPHICS...

proc corr data=sashelp.class plots=matrix;

# SUGGESTION, START SLOWLY AND SIMPLY

# USE SAME APPROACH AS WHEN LEARNING OTHER
NEW THINGS IN SAS

WHAT HAPPENS WHEN SAS MAKES ALL THE
DECISIONS (WHAT IS THE DEFAULT BEAHAVIOR)

HOW DO I ADD STATEMENTS TO MODIFY THE
DEFAULT BEHAVIOR

# PLOT GPLOT... BUILD A PLOT STEP-BY-STEP

HEALTH CARE EXPENDITURES IN THE UNITED
STATES FROM 1970 THROUGH 2000

CREATE A PLOT THAT SHOWS THE TREND IN
EXPENDITURES...

1/ PLOT TOTAL EXPENDITURES BY YEAR

2/ ADD FEDERAL GOVERNMENT EXPENDITURES

MINIMUM: DATA SET, PLOT STATEMENT

* #1: PLOT NH_TOT VERSUS YEAR 
      LET SAS DESIGN THE PLOT;

proc gplot data=t112;
plot nh_tot*year;
run;
quit;

NOTICE ...

Y*X

SCATTER
WITH +

AXIS SCALES
AXIS LABELS

FRAMED

FONT/SIZE

GLOBAL: SYMBOL

* #2: CHANGE THE PLOTTING SYMBOL 
      AND JOIN THE POINTS; 

symbol value=dot interpol=join;

proc gplot data=t112;
plot nh_tot*year;
run;
quit;

NEW
PLOTTING
SYMBOL

POINTS
CONNECTED
WITH LINE
SEGMENTS

SYMBOL STATEMENT ...

Syntax...

SYMBOL<1...255>  <COLOR=symbol-color> <MODE=EXCLUDE | INCLUDE>
<REPEAT=number-of-times> <STEP=distance<units>>
<appearance-option(s)> <interpolation-option> <SINGULAR=n>;

appearance-options can be one or more of these:
 BWIDTH=box-width
 CI=line-color
 CO=color
 CV=value-color
 FONT=font
 HEIGHT=symbol-height<units>
 LINE=line-type
 POINTLABEL<=(label-description(s)) | NONE>
 VALUE=special-symbol | text-string | NONE
 WIDTH=thickness-factor

interpolation-option can be one of these:

general methods
 INTERPOL=JOIN
 INTERPOL=map/plot-pattern
 INTERPOL=NEEDLE
 INTERPOL=NONE
 INTERPOL=STEP<placement><J><S>

high-low interpolation methods
 INTERPOL=BOX<option(s)><00...25>
 INTERPOL=HILO<C><option(s)>
 INTERPOL=STD<1 | 2 | 3><variance><option(s)>

regression interpolation methods
 INTERPOL=R<type><0><CLM | CLI<50...99>>

spline interpolation methods
INTERPOL=L<degree><P><S>
INTERPOL=SM<nn><P><S> 
INTERPOL=SPLINE<P><S>

GLOBAL: TITLE, FOOTNOTE (NOT SPECIFIC TO
SAS/GRAPH)

* #3: ADD A TITLE AND FOOTNOTE;

symbol v=dot i=join;

title "NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote "Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year;
run;
quit;

BOTH TITLE
AND
FOOTNOTE 
ARE
CENTERED

TITLE FONT
IS LARGER
THAN THAT
OF THE
FOOTNOTE

PLOT SIZE IS
SMALLER

NOTICE:  YOU CAN PUT TITLES/FOOTNOTES ON LEFT & RIGHT

GOPTIONS: CHANGE DEFAULTS

* #4: USE A GOPTIONS STATEMENT 
      TO CONTROL THE  TEXT FONT AND HEIGHT
      (SYMBOL SIZE AFFECTED);

goptions ftext='Arial' htext=2 gunit=pct;

symbol v=dot i=join;

title "NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote "Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year;
run;
quit;

WHAT DOES H=2 MEAN 

ALL OUTPUT
DESTINATIONS HAVE
CHARACTER CELLS

PROC GTESTIT;
RUN;

UPPER RIGHT

R=65 C=137
SO ... H=2/65

(CELLS, CM, IN, PCT, PT) 

GLOBAL: OVERRIDE GOPTIONS

* #5: CHANGE THE HEIGHT OF THE TITLE
      RIGHT JUSTIFY THE FOOTNOTE;

goptions ftext='Arial' htext=2 gunit=pct;

symbol v=dot i=join;

title height=4 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote justify=right 
"Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year;
run;
quit;

GLOBAL: SYMBOL, TITLE, FOOTNOTE USED
AUTOMATICALLY

AXIS NOT USED AUTOMATICALLY, MUST
SPECIFY USE

* #6: CHANGE THE Y-AXIS LABEL
      BOTH THE CONTENT AND ANGLE;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct;

symbol v=dot i=join;
axis label=(angle=90 "AMOUNT (IN BILLIONS)");

title1 h=4 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote1 j=right 
"Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year/vaxis=axis;
format nh_tot comma.;
run;
quit;

* #7: ADDITIONAL CHANGES OF BOTH THE X AND Y AXES;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct;

symbol v=dot i=join;
axis1 label=(angle=90 "AMOUNT (IN BILLIONS)")  
      minor=(n=3);
axis2 order=(1970 to 2000 by 5) minor=(n=4);

title h=4 "NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote justify=right "Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year/vaxis=axis1 haxis=axis2;
format nh_tot comma.;
run;
quit;

* #8: ADD COLOR;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct ctext=green;

symbol v=dot i=join c=blue h=2.5;
axis1 label=(angle=90 "AMOUNT (IN BILLIONS)") 
      minor=(n=3) color=blue;
axis2 order=(1970 to 2000 by 5) minor=(n=4) color=blue;

title1 h=4 "NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
footnote1 j=r "Source: Health-United States-2003";

proc gplot data=t112;
plot nh_tot*year/vaxis=axis1 haxis=axis2;
format nh_tot comma.;
run;
quit;

* #9: LEAVE ALL AXES TEXT GREEN;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct ctext=green;

symbol v=dot i=join c=blue h=2.5;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
      minor=(n=3); 
axis2 order=(1970 to 2000 by 5) 
      minor=(n=4) offset=(2,2); 

title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=r ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot nh_tot*year/vaxis=axis1 haxis=axis2 caxis=blue;
format nh_tot comma.;
run;
quit;

* #10: ADD ANOTHER LINE TO THE PLOT;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct ctext=green;
symbol v=dot i=join c=blue h=2.5;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
      minor=(n=3); 
axis2 order=(1970 to 2000 by 5) minor=(n=4) offset=(2,2); 
title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=r ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
      vaxis=axis1 haxis=axis2 caxis=blue;
format nh_tot comma.;
run;
quit;

* #11: ADD ANOTHER SYMBOL STATEMENT;
goptions reset=all;
options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct ctext=green;
symbol1 v=dot i=join h=2.5 l=1;
symbol2 v=dot i=join h=2.5 l=3;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
       minor=(n=3); 
axis2 order=(1970 to 2000 by 5) minor=(n=4) offset=(2,2); 
title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=r ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
      vaxis=axis1 haxis=axis2 caxis=blue;
format nh_tot comma.;
run;
quit;

* #12: USE GOPTIONS AND THE CYSMBOL OPTION;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct 
         ctext=green csymbol=blue;

symbol1 v=dot i=join h=2.5 l=1;
symbol2 v=dot i=join h=2.5 l=3;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
      minor=(n=3); 
axis2 order=(1970 to 2000 by 5) minor=(n=4) offset=(2,2); 

title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=right ls=1 
"Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
     vaxis=axis1 haxis=axis2 caxis=blue;
format nh_tot comma.;
run;
quit;

* #13: ADD A LEGEND;

options validvarname=upcase;
goptions ftext='Arial' htext=2 gunit=pct 
         ctext=green csymbol=blue;
symbol1 v=dot i=join h=2.5 l=1;
symbol2 v=dot i=join h=2.5 l=3;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
      minor=(n=3); 
axis2 order=(1970 to 2000 by 5) minor=(n=4) offset=(2,2); 
title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=r ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
     vaxis=axis1 haxis=axis2 caxis=blue legend;
format nh_tot comma.;
run; quit;

* #14: MODIFY THE DEFAULT LEGEND;

<NO CHANGES IN OPTIONS, GOPTIONS, SYMBOLS, AXES>

legend1 label=none 
        value=(j=left "TOTAL" j=left "FEDERAL GOVT") 
        mode=protect position=(top inside left) 
        cborder=blue cshadow=blue 
        across=1 shape=line(10);

title1 h=4 f='Arial/bo' ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=r ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
     vaxis=axis1 haxis=axis2 caxis=blue legend=legend1;
format nh_tot comma.;
run; quit;

# ANNOTATE DATA SET

DATA SET THAT CONTAINS VARIABLES THAT ARE
"UNDERSTOOD" BY SAS/GRAPH PROCEDURES
(ARE YOU FAMILIAR WITH CNTLIN DATA SETS?)

ANALOGOUS TO EDITING THE OUTPUT PRODUCED
BY PROCEDURES WITH A GRAPHICS EDITOR,
EXCEPT ALL INSTRUCTIONS ARE WRITTEN

WHERE, WHAT, HOW

* ANNOTATE DATA SET WITH SPECIAL VARIABLE NAMES;

data labels;
retain xsys ysys '2' function 'label' 
       position '1'  style '"Tahoma/bo" '
       color 'blue'  cborder 'blue';

* find last observation in data set t112;
set t112 end=last;

if last then do;
   text=' FEDERAL GOVERNMENT '; x=year; y=fg_tot;
output;
   text=' TOTAL '             ; x=year; y=nh_tot;
output;
end;
run;

* ANNOTATE DATA SET WITH SPECIAL VARIABLE NAMES;

data labels;
retain xsys ysys '2' function 'label' 
       position '1'  style '"Tahoma/bo" '
       color 'blue'  cborder 'blue';

* find last observation in data set t112;
set t112 nobs=lastrec point=lastrec;

text=' FEDERAL GOVERNMENT '; x=year; y=fg_tot; output;
text=' TOTAL '             ; x=year; y=nh_tot; output;

stop;
run;

* #16 USE ANNOTATION TO LABEL THE LINES;

options validvarname=upcase;
goptions ftext='Tahoma/bo' htext=2 gunit=pct 
         ctext=green csymbol=blue;

symbol1 f=marker v='C' i=join h=1.25;
symbol2 f=marker v='U' i=join h=1.25;
axis1 label=(angle=90 rotate=0 "AMOUNT (IN BILLIONS)")
      minor=(n=3); 
axis2 order=(1970 to 2000 by 5) minor=(n=4) offset=(2,2); 

title1 h=4 ls=1 
"NATIONAL HEALTH CARE EXPENDITURES: 1970-2000";
title2 h=2 a=90; title3 h=2 a=270;
footnote j=right ls=1 "Source: Health-United States-2003  ";

proc gplot data=t112;
plot (nh_tot fg_tot)*year/overlay 
      vaxis=axis1 haxis=axis2 caxis=blue annotate=labels; 
format nh_tot comma.;
run; quit;

P AFTER ALL THIS, WHY USE SAS/GRAPH

DATA IN SAS DATA SETS, STAY WITHIN SAS TO
PRODUCE GRAPHICS

GOPTIONS, PROCEDURE OPTIONS, ANNOTATE
DATA SETS CAN PRODUCE ANY TYPE OF OUTPUT

REPETITIVE PRODUCTION OF GRAPHICS CAN BE
AUTOMATED USING DATA STEP PROGRAMMING
AND MACROS



SAS/GRAPH® 101

51

P COMMUNITY HEALTH DATA SET

100+ BAR CHART+MAP
6,000+ PLOT+TABLE
UPDATED 'REGULARLY'

# PAPER ON WEB SITE HAS HYPERLINKS TO OTHER
SAS/GRAPH RELATED PAPERS

# ALL SAS CODE FROM PAPER ON THE WEB SITE

# ROBERT ALLISON'S "UNOFFICIAL" WEB SITE WITH
EXAMPLES AND SAS CODE

http://www.robslink.com/SAS/Home.htm

http://www.robslink.com/SAS/Home.htm



